2017.12.7

Mizarによる微分幾何形式化の準備状況
(陰関数定理)

TPP2017 資料

師玉康成(信州大学工学部電子情報システム工学科)

1　概　要

この数年，証明支援系を利用して数学定理の証明を計算機により厳密検証する事例が相次いで報告された．しかし，多様体論などの幾何学は直観に強く依存した分野であるため，証明支援系による形式化が遅れている．筆者らは，ストークスの定理を最終目標に設定し，証明支援系 Mizar を用いて滑らかな多様体およびその周辺の形式化を行なっている．これまでに滑らかな多様体を証明支援系によって形式化した事例はなく，滑らかな多様体はそれ自体が興味深い研究対象であるとともに，現代数学の基本的な道具でもある．この報告ではその計画概要と現状について説明する．

2　背　景

ここ数年，Feit-Thompson の定理，ケプラー予想など，証明支援系を用いて証明を計算機によって厳密検証する事例が相次いで報告された．現時点では，表 1 のように形式化には多大な労力が必要とされるが，今後の技術発展により，証明支援系によって重要な数学論文が査読時に厳密検証される時代が到来するものと考えられる．

表 1．形式化された代表的な数学定理と，形式化に費やされたリソース

定理	証明支援系	終了年	期間(年)	人数	行数
四色問題*1	Coq	2004	6	1	60,000
Jordan 曲線定理*2	Mizar	2005	14	16	200,000
Feit-Thompson の定理*3	Coq	2012	6	15	170,000
ケプラー予想*4	HOL light	2014	12	16	500,000

*1 四色問題:平面上のいかなる地図も、隣接する領域が異なる色になるように塗り分けるには 4 色あれば十分
*2 Jordan 曲線定理：平面に置かれた自己交差を持たないどんな閉曲線（輪っか）も平面を「内側」と「外側」に分ける
*3 Feit-Thompson の定理：有限群に関する定理（奇数位数の有限群はすべて可解である。）
*4 ケプラー予想：三次元ユークリッド空間における球充填に関する数学的な予想

Freek Wiedijk によると，1999 年に Jack and Paul Abad によって発表された「The 100 Greatest Theorems」のうち，すでに 92 の定理については証明支援系による形式化が完了し，計算機によって厳密検証されたとのことである．表 2 では，このリストから形式化が完了した定理を抜粋した．

表 2. The 100 Greatest Theorems で証明支援系による厳密証明が完了したもの (一部抜粋)

代数学の基本定理，素数定理，ゲーデルの不完全性定理，平方剰余の相互法則，オイラー標数，リウヴィルの定理と超越数の構築，四平方の定理，四色問題，テイラーの定理，ブラウワーの不動点定理，ベズーの定理

このように，代数学・解析学を始めとした分野については，多くの定理が証明支援系によって形式化されている．ところが，これらの分野と比較すると，多様体論など現代数学において重要な幾何概念の証明支援系による形式化は，ほとんど手付かずのままである．これは，幾何が直観（イメージ）に強く依存した分野であるため，通常証明と（厳密な）形式証明との乖離が大きく，証明支援系による形式化が避けられてきたためと考えられる．

千葉大・久我健一らの研究グループは証明支援系 Coq を用いてホモトピー型理論（HoTT）の形式化を進めている．

その一方で, 3 次元ポアンカレ予想の解決や一般相対論などの事例を挙げるまでもなく，現代の数学や物理学において微分幾何は必要不可欠な土台であり，形式化においても避けて通ることはできない. 筆者らの知る範囲では 2017 年現在, 証明支援系による滑らかな多様体の形式化事例は見られない.

３　計画概要

筆者らは証明支援系 Mizar を用いて，ストークスの定理の厳密証明を最終目標に設定し, 滑らかな多様体とその周辺の形式化を行なっている. これにより, 以下の概念および定理が形式化される見通しである.

・逆函数定理と陰函数定理

・滑らかな多様体の定義と諸性質（多様体上の函数，多様体間の写像）

・接ベクトル空間，部分多様体，多様体の向き付け，境界のある多様体， 1 の分割

・微分形式に関する諸性質（座標不変性，引き戻し，制限，外微分，積分など）

解析学系のライブラリ作成をもとに目的の幾何学のライブラリ作成を目指しているがそれぞれの概念、定理の関係は以下の通りである



解析学からの準備として，逆函数定理と陰函数定理，隆起函数などを形式化する．その準備のもとに，多様体論の形式化を進め，滑らかな多様体，接ベクトル空間，部分多様体，多様体の向き付けと境界，１の分割，微分形式や外微分とその諸性質，多様体上での積分，ストークスの定理の順に形式化を進める．

幾何学は人間の直観が入り込みやすい分野であるため，教科書通りに形式化すればよい，という具合にはいかない．特に級数や幾何的なイメージを伴う概念の形式化には注意が必要で，行き詰まった際は担当にはこだわらず臨機応変に対応するしかない．

４　形式化の作業順序

今年から作業は開始されたが,以下のような順序で行うことを予定している。作業の分担予定は暫定的なものである。

(1) の陰関数定理と逆関数定理については既に筆者を含む日本側の形式化作業で粗完成したが,以降の作業はポーランド・Bialystok 大学の研究協力者 7 名を加えて推進する。

また，ストークスの定理の形式化には多重積分の形式化が必要であり,多重積分を累次積分で表すことをルベーグ積分に拡張した Fubini の定理も必要であるが,これについては既に日本側のグループの遠藤が形式化を略完了している。[1]

(1)　**逆函数定理と陰函数定理（担当：日本）**

函数定理と陰函数定理は，バナッハの不動点定理を用いた Jürgen Jost の[2]の証明にもとづいて形式化を行なう．すでにバナッハの不動点定理など必要な道具は揃えた。

⑵　隆起函数（担当：ポーランド）

１の分割を形式化するために, n 次元ユークリッド空間上のコンパクトな台をもつ滑らかな函数を構成する．現時点では，級数の扱いは形式化における厄介事の一つであり，テイラー展開による上からの評価では，若干の困難が予想される．

以上の準備のもと,多様体論を本格的に形式化する．以降は，Loring W. Tu による多様体論の標準的な教科書[3]を参考として進める．

⑶滑らかな多様体（担当：日本）

すでに位相多様体については形式化が完了しているため，アトラスが滑らかに両立することの定義を組み込めばよい．また，多様体上の函数および，多様体間の写像や微分同相についてもここで形式化する．

⑷　接ベクトル空間（担当：ポーランド）

接ベクトル空間は部分多様体や曲面の向き付けなどで必要となる．ユークリッド空間におけるヤコビ行列と連鎖律などの諸性質を形式化するとともに接ベクトル空間を定義した上で，多様体間の写像へ拡張し，座標非依存であることを形式化する．

また，多様体間の写像から導出される微分（接ベクトル空間の写像）および，正則点，臨界点などの定義を行なう．

⑸　部分多様体（担当：日本）

１で形式化した逆函数定理を用いて，多様体 M, N の間の写像 f : M -> N が埋め込みである場合に M と f(M)が微分同相であることを形式化する．

⑹　多様体の向き付け（担当：日本）

多様体の向き付け可能性の定義および，向き付け可能な場合の向き付けの数は 2 通りであることを形式化する．多様体の向き付けは微分形式の存在証明に必要となる．

⑺　多様体の境界（担当：ポーランド）

境界付き多様体の定義，定義が well-defined であることを形式化する．さらに，境界付きの n 次元多様体の境界は，境界のない n-1 次元多様体であることを形式化する．また，多様体の向き付けの境界への引き継ぎについても形式化する．これらはストークスの定理において微分形式を積分する際に必要となる．

⑻　１の分割（担当：日本）

(3)で形式化した隆起函数を用いて，１の分割の定義と，滑らかなコンパクト多様体上での存在について形式化する．１の分割は，滑らかな多様体を論ずる上での基本的な道具であり，ホイットニーの埋め込み定理やリーマン計量の存在証明にも通じるため，今回形式化できれば意義は大きい．

(9)　微分形式（担当：日本）

まず n 次元ユークリッド空間上での微分形式の諸性質（座標不変性，引き戻し，境界への制限，連鎖律，行列式）を形式化したのち，滑らかな多様体上での微分形式を形式化する．また，外微分およびその諸性質（外微分の存在，座標不変性，引き戻しとの関係）についても，ここで形式化を行なう．

(10)　多様体上での積分（担当：ポーランド）

前年までの準備を踏まえ，多様体上での積分を定義する．すでにユークリッド空間上でのリーマン積分は形式化されている．１の分割のとり方などによらず一意に定まることや，引き戻しとの関係性を形式化する．

(11)　ストークスの定理（担当：日本）

本研究の最終目標であるストークスの定理の形式化を完成させる．これにより，電磁気学やコホモロジー論への形式化の道が通じることとなる．時間が許せば，本結果の２次元の場合であるグリーンの定理についても形式化をおこなう．グリーンの定理が得られれば，すぐにコーシーの積分定理が従うため，複素解析の形式化への道も開けることとなる．

以上の作業順序と役割分担を図式化すると以下のようになる。



解析学からの準備（平成２９年度）
バナッハの不動点定理
フビニの定理
1. 逆函数定理，陰函数定理
（担当：日本）
2. 隆起函数
$$\phi(x) = \begin{cases} \exp\left(-\frac{1}{1-x}\right) for \ |x| < 1 \\ 0 \qquad otherwize \end{cases}$$
・滑らかであることの証明
（担当：ポーランド）
多様体論（平成３０年度以降）
3. 滑らかな多様体
・アトラスの両立
・多様体上の函数
・多様体間の写像と同相
（担当：日本）
4. 接ベクトル空間
・座標不変性
・ヤコビ行列と連鎖律
・多様体間写像の微分
・正則点，臨界点
（担当：ポーランド）
5. 部分多様体
・埋め込みと微分同相
（担当：日本）
7. 多様体の境界
・境界つき多様体の定義
・境界が多様体であること
・向き付けの境界への引継
（担当：ポーランド）
6. 多様体の向き付け
・向き付け可能性の定義
・向き付けの数
（担当：日本）
9. 微分形式
・座標不変性
・引き戻し(Pullback)
・境界への制限
・連鎖律
・行列式
・外微分と諸性質
（担当：日本）
8. １の分割
・１の分割の定義
・滑らかなコンパクト多様体における存在証明
（担当：日本）
10. 多様体上での積分
・Well-definedであることの証明
・引き戻しとの関係性
（担当：ポーランド）
11. ストークスの定理
$$\int_M d\omega = \int_{\partial M} \omega$$
（担当：日本）

5　現状の進捗状況

４節で述べたように(1)の陰関数定理,逆関数定理及び多重積の形式化に必要な Fubini の定理については既に筆者を含む日本側の作業で粗完成している。Fubini の定理については以下のような形式化を行った[1],[MML: MEASUR11]が，本報告では説明を省略し,この項では陰関数定理の形式化について概説する。

```
theorem :: MEASUR11:118
for X1,X2 be non empty set, S1 be SigmaField of X1, S2 be SigmaField of X2,
  M1 be sigma_Measure of S1, M2 be sigma_Measure of S2,
  E be Element of sigma measurable_rectangles(S1,S2)
st M1 is sigma_finite & M2 is sigma_finite
holds Integral(M2,(X-vol(E,M1))) = (product_sigma_Measure(M1,M2)).E;
```

筆者らは以下のような実ノルム線形空間上の陰関数定理とその証明の形式化を行った。

陰関数定理(シュヴァルツ解析学２　微分法　東京図書　p.104)

E,F,G をノルム空間とし ,$f(x,y)$を$E\times F$の開集合Ωから Gへの連続写像とする。

1) A,BをそれぞれE,Fの開集合で$A\times B\subseteq\Omega$とし,gはAからBへの連続写像ですべての$x\in A$について

$$f(x,g(x))=c$$

を満たすものとする。　Aのすべてのxに対し$\frac{\partial}{\partial y}f\big(x,g(x)\big)$が$\mathcal{L}(F,G)$の可逆元ならば$g$は$A$から$B$への連続可導な写像である

2) (a,b)がΩの点，$f(a,b)=c$とする。$\frac{\partial}{\partial y}f(a,b)$が$\mathcal{L}(F,G)$の可逆元で,F が完備ならば a,bを含むE,Fの開集合をA,B, $A\times B\subseteq\Omega$を適当にとってすべての$x\in A$についてyについての方程式

$$f(x,y)=c$$

*がBにおいて唯一の解をもつようにでき,このように定義された関数$y=g(x)$はAからBへ*の連続可導な写像である。さらにgの導関数は

$$g'(x)=-\left\{\frac{\partial}{\partial y}f\big(x,g(x)\big)\right\}^{-1}\frac{\partial}{\partial x}f(x,g(x))$$

で与えられる。

注：上記の教科書シュヴァルツ解析学 2 ではノルムアフィン空間で記述されているがここではノルム空間のものに変えてある

実線形ノルム空間は
Mizar Mathematical Library:
http://mizar.uwb.edu.pl/version/current/html/
(以降 MML)に以下のように形式化されている。[MML: NORMSP_1]

```
definition
  struct(RLSStruct, N-ZeroStr) NORMSTR (# carrier -> set,   ZeroF -> Element of the carrier,
   addF -> BinOp of the carrier, Mult -> Function of [:REAL, the carrier:], the carrier, normF -> Function of the carrier, REAL #);
end;
```

実線形ノルム空間の構造は実線形空間 RLSStruct とノルム N-ZeroStr の構造から合成される。
ノルムについての斉次性、劣化加法性は次のように属性として定義されている。

```
definition
  let IT be non empty NORMSTR;
  attr IT is RealNormSpace-like means  :: NORMSP_1:def 1
  for x, y being Point of IT, a holds ||.a * x.|| = |.a.| * ||.x.|| & ||.x + y  .|| <= ||.x.|| + ||.y.||;
end;
```

実線形ノルム空間の構造に上のノルムについての斉次性、劣化加法性や加法、スカラ倍に関する交換、結合則、分配則などを課したものとして実線形ノルム空間が定義される

```
definition
  mode RealNormSpace is reflexive discerning RealNormSpace-like
  vector-distributive scalar-distributive scalar-associative scalar-unital
  Abelian add-associative right_zeroed right_complementable
   non empty NORMSTR;
end;
```

ノルム線形空間 S からノルム線形空間 T への有界線形写像全体もノルム線形空間であり、R_NormSpace_of_BoundedLinearOperators(S,T)で表すがこれは以下のように形式化されている。R_VectorSpace_of_LinearOperators(S,T)は実線形空間 S,から実線形空間 T への写像全体からなる線形空間で
, R_NormSpace_of_BoundedLinearOperators(S,T)はその部分線形空間として導入され加法、スカラ倍を制限した演算と有界線形作用素のノルムを持ってい

るものとして形式化されている。[MML: LOPBAN_1]

```
definition
  let S,T be RealNormSpace;
  func  R_NormSpace_of_BoundedLinearOperators(S,T)  ->  non  empty  NORMSTR
equals
:: LOPBAN_1:def 14
  NORMSTR (# BoundedLinearOperators(S,T), Zero_(BoundedLinearOperators(S,T),
    R_VectorSpace_of_LinearOperators(S,T)), Add_(BoundedLinearOperators(S,T),
    R_VectorSpace_of_LinearOperators(S,T)), Mult_(BoundedLinearOperators(S,T),
    R_VectorSpace_of_LinearOperators(S,T)), BoundedLinearOperatorsNorm(S,T) #);
end;
```

ノルム線形空間 S からノルム線形空間 T へ写像 f の微分は, S から T への有界線形写像として定義される。　これは以下のように形式化されている。　[MML: NDIFF_1]

```
definition
  let S,T be RealNormSpace;
  let f be PartFunc of S,T;
  let x0 be Point of S;
  pred f is_differentiable_in x0 means  :: NDIFF_1:def 6
  ex N being Neighbourhood of x0
st N c= dom f & ex L be Point of R_NormSpace_of_BoundedLinearOperators(S,T),
          R be RestFunc of S,T   st for x be Point of S st x in N holds
    f/.x - f/.x0= L.   (x-x0) + R/.(x-x0);
end;
definition
  let S,T be RealNormSpace;
  let f be PartFunc of S,T;
  let x0 be Point of S;
  assume
 f is_differentiable_in x0;
  func   diff(f,x0)   ->   Point   of   R_NormSpace_of_BoundedLinearOperators(S,T)
means::NDIFF_1:def 7
  ex N being Neighbourhood of x0 st N c= dom f & ex R st for x be Point of   S
    st x in N holds f/.x-f/.x0 = it.(x-x0) + R/.(x-x0);
end;
```

ここで R は 残項　RestFunc of S,T であるが,これは

$$\lim_{||y||\to 0} \frac{\left||R(y)|\right|}{||y||} = 0$$

を満たすものとして以下のように形式化されている

```
definition
  let S,T be RealNormSpace;
  let IT be PartFunc of S,T;
  attr IT is RestFunc-like means
:: NDIFF_1:def 5
  IT is total & for h st h is non-zero holds (||.h.||")(#)(IT
  /*h) is convergent & lim ((||.h.||")(#)(IT/*h)) = 0.T;
end;
```

実ノルム線形空間 X,Y,W について X と Y の直積空間 X×Y から W への写像fの偏微分

$$\frac{\partial}{\partial x} f(x,y), \quad \frac{\partial}{\partial y} f(x,y)$$

も微分と同様にそれぞれ X から W, Y から W への有界線形写像として定義されるが

以下のように形式化されている。[MML: NDIFF_ 7]

```
definition
  let X, Y, W be RealNormSpace,
  z be Point of [:X,Y:],
  f be PartFunc of [:X,Y:], W;
  pred f is_partial_differentiable_in`1 z means   :: NDIFF_7:def 4
    f*reproj1(z) is_differentiable_in z`1;
  pred f is_partial_differentiable_in`2 z means   :: NDIFF_7:def 5
    f*reproj2(z) is_differentiable_in z`2;
end;
definition
  let X, Y, W be RealNormSpace,
  z be Point of [:X,Y:],
  f be PartFunc of [:X,Y:], W;
  func partdiff`1(f,z) -> Point of R_NormSpace_of_BoundedLinearOperators(X,W)
```

```
  equals  :: NDIFF_7:def 6
  diff(f*reproj1(z),z`1);

  func partdiff`2(f,z) -> Point of R_NormSpace_of_BoundedLinearOperators(Y,W)
  equals  :: NDIFF_7:def 7
  diff(f*reproj2(z),z`2);
end;
```

ここで z`1 =x のとき reproj1(z) .y=[x,y]であり, z`2 =y のとき reproj2(z) .x=[x,y] となる関数である。 また　z=[x,y]　のとき z`1 =x, z`2 =y　となる射影である。

この定理の証明の形式化は, バナッハの不動点定理を用いた Jürgen Jost の[2]の証明にもとづいて行なった。下記のようにバナッハの不動点定理など必要な道具は揃えながら作業を進めた。上記の教科書も同様の方法によっている。

まず,関数

$$f_1(x,y) = f(x,y+b) - c$$

$$\Psi(x,y) = y - f_1(x,y)$$

を導入して点(a,b)の近傍での同値な関係式

$f(x,y) = c \quad \Longleftrightarrow \quad \Psi(x,y) = y$

を使って, xに対するyについての方程式$f(x,y) = c$の解の存在性を,パラメータxを伴った関数$\Psi(x,y)$の不動点問題$\Psi(x,y) = y$に帰着させる。

そのため不動点定理が必要になる。筆者らは以下のように縮小写像原理による不動点定理を形式化した。縮小写像には不動点の存在性と一意性が成り立っている。

```
theorem :: NDIFF_8:32
  for X be RealBanachSpace,
      S be non empty Subset of X,
      f be PartFunc of X,X
  st S is closed & dom f = S & rng f c= S
   & ex k be Real    st 0 < k & k < 1
      & for x,y be Point of X st x in S & y in S
         holds ||. f/.x - f/.y .|| <= k * ||.x-y.||
  holds
    (ex x0 be Point of X st x0 in S & f.x0 = x0)
   &(for x0,y0 be Point of X st x0 in S & y0 in S & f.x0 = x0 & f.y0 = y0
      holds x0 = y0);
```

さらにこれを用いてパラメータxを伴った関数$\Psi(x,y)$の不動点問題$\Psi(x,y) =$

yについてパラメータxを不動点 y に対応させる関数の一意存在性とその連続性に関する命題を形式化した。

```
theorem :: NDIFF_8:34
  for E be RealNormSpace,
      F be RealBanachSpace,
      A be non empty Subset of E,
      B be non empty Subset of F,
      Fai be PartFunc of [:E,F:],F
   st B is closed
    & [:A,B:] c= dom Fai
    & (for x be Point of E, y be Point of F
          st x in A & y in B holds Fai.(x,y) in B )
    & (for y be Point of F st y in B holds
        for x0 be Point of E st x0 in A
        for e be Real st 0 < e
        ex d be Real
          st 0 < d
           & for x1 be Point of E
                st x1 in A & ||.x1-x0.|| < d
              holds ||. Fai/.[x1,y] - Fai/.[x0,y] .|| < e)
    & ex k be Real
        st 0 < k & k < 1
         & (for x be Point of E st x in A holds
             for y1,y2 be Point of F st y1 in B & y2 in B
             holds ||. Fai/.[x,y1] - Fai/.[x,y2] .|| <= k * ||.y1-y2.||)
  holds
  (ex g be PartFunc of E,F
    st g is_continuous_on A
     & dom g = A & rng g c=  B
     & for x be Point of E st x in A
       holds Fai.(x,g.x) = g.x)
     &(for g1,g2 be  PartFunc of E,F
        st dom g1 = A & rng g1 c= B
         & dom g2 = A & rng g2 c= B
```

```
    &(for x be Point of E st x in A
      holds Fai.(x,g1.x) = g1.x)
    &(for x be Point of E st x in A
      holds Fai.(x,g2.x) = g2.x)
  holds g1=g2);
```

点 (a, b) の近傍内上での関数 g の可動性にはその近傍内の点 (x, y) での $\frac{\partial}{\partial y}f\big(x,g(x)\big)$の可逆性が必要になる。$\frac{\partial}{\partial y}f\big(x,g(x)\big)$はノルム空間の間の有界線形作用素であるから有界線形作用素の可逆性に関する定理を使う。$\frac{\partial}{\partial y}f(a,b)$の可動性から (a, b) から (x, y) への摂動した$\frac{\partial}{\partial y}f\big(x,g(x)\big)$の可動性を導く。

```
theorem :: LOPBAN16:18
 for X,Y be non trivial RealBanachSpace,
       u,v be Point of  R_NormSpace_of_BoundedLinearOperators(X,Y)
   st  u is invertible    & ||.v-u.|| <  1/||.Inv u .||    holds
         v is invertible
        & ||.Inv v.|| <=1/ ( 1/||.Inv (u).|| - ||.v-u.|| )
        & ex w be Point of
          R_Normed_Algebra_of_BoundedLinearOperators X,
          s,I be Point of   R_NormSpace_of_BoundedLinearOperators(X,X)
          st w=(Inv u) * (v-u)
             & s = w & I = id the carrier of X
             & ||.s.|| < 1    & (-w) GeoSeq is norm_summable
             & I+s is invertible   & ||.Inv (I+s).|| <=1/ ( 1 - ||.s.|| )
             & Inv(I+s) = Sum ( (-w) GeoSeq )    & Inv v = Inv(I+s) * (Inv u);
```

Inv u, Inv v はそれぞれ作用素 u,v の逆作用素u^{-1}, v^{-1}を表している。作用素 u の可逆で逆作用素u^{-1}が存在し，作用素ノルムについて$||.u\text{-}v.|| < 1$ が成り立っているとき v が可逆で逆作用素 v^{-1}存在することを表している。またそのv^{-1}が関数級数の和で求められることを表している。

以上の準備のもとに陰関数定理を以下のように形式化した。

```
theorem :: NDIFF_9:21
  for E be RealNormSpace,
      G,F be non trivial RealBanachSpace,
      Z be Subset of [:E,F:],
      f be PartFunc of [:E,F:], G,
      a be Point of E,
      b be Point of F,
      c be Point of G,
      z be Point of [:E,F:]
     st
     Z is open    & dom f = Z
   & f is_differentiable_on Z
   & f `| Z is_continuous_on Z
   & [a,b] in Z & f.(a,b) = c
   & z = [a,b]   & partdiff`2(f,z) is invertible            ∂/∂y f(a,b)が可逆
holds
ex r1,r2 be Real    st 0 < r1 & 0 < r2
  & [:Ball(a,r1),cl_Ball(b,r2):] c= Z                         開球 Ball(a,r1)の任意の元 x
  & ( for x be Point of E st x in Ball(a,r1) holds           について y に関する
           ex   y be Point of F                               方程式 f.(x,y) = c は唯一の解を
             st y in Ball(b,r2) &   f.(x,y) = c )            もつ
 &  ( for x be Point of E st x in Ball(a,r1) holds
          for y1,y2 be Point of F
           st   y1 in Ball(b,r2) & y2 in Ball(b,r2)
        & f.(x,y1) =c & f.(x,y2)= c    holds y1=y2
& (   ex g be PartFunc of E,F
st dom g = Ball(a,r1)
 & rng g c= Ball(b,r2)
 & g is_continuous_on Ball(a,r1)                    開球 Ball(a,r1)の任意の元 x について
 & g.a = b                                          f(x,g(x))=c を満たす
 & ( for x be Point of E st x in Ball(a,r1)         連続関数 g が存在する
           holds f.(x,g.x) = c )
&   g is_differentiable_on Ball(a,r1)                 g は連続可導な写像
&   g `| Ball(a,r1) is_continuous_on   Ball(a,r1)
&   (   for x be Point of E,
              z be Point of [:E,F:]
```

```
              st x in Ball(a,r1) & z =[x,g.x]
               holds
                diff(g,x) = - ( Inv partdiff`2(f,z))
                     * partdiff`1(f,z) )
```

$$\boxed{g'(x) = -\left\{\frac{\partial}{\partial y} f\big(x, g(x)\big)\right\}^{-1} \frac{\partial}{\partial x} f(x, g(x))}$$

```
  & ( for x be Point of E,
            z be Point of [:E,F:]
          st x in Ball(a,r1) & z =[x,g.x]
               holds
              partdiff`2(f,z) is invertible ) )
```

６　終わりに

以上、Mizar による微分幾何の形式化の準備状況を概説した。当面の目標はストークス定理などの形式化を目指しているが、本報告では，バナッハの不動点定理を用いた Jürgen Jost の[2]の証明にもとづいた陰関数定理の形式化について説明した。冒頭に述べたように多様体論など現代数学において重要な幾何概念の証明支援系による形式化は，ほとんど手付かずのままである．幾何が直観（イメージ）に強く依存した分野であるため，通常証明と（厳密な）形式証明との乖離が大きく，証明支援系による形式化が難しい．しかしながら現代の数学や物理学において微分幾何は必要不可欠な土台であり，形式化においても避けて通ることはできない．筆者らの試みがこの状況の改善の一助になることを期待する。

参考文献

[1] Noboru Endo:Fubini's Theorem on Measure, FORMALIZED MATHEMATICS,25,1, pp. 1–29, 2017

[2] Jost, Jürgen. Postmodern analysis. Springer Science & Business Media, 2006.

[3] Tu, Loring W. An introduction to manifolds. Springer Science & Business Media, 2010.

MML(Mizar Mathematical Library: http://mizar.uwb.edu.pl/version/current/html/)

NORMSP_1: http://mizar.uwb.edu.pl/version/current/html/normsp_1.html
NDIFF_1: http://mizar.uwb.edu.pl/version/current/html/ndiff_1.html
NDIFF_7: http://mizar.uwb.edu.pl/version/current/html/ndiff_7.html
MEASUR11: http://mizar.uwb.edu.pl/version/current/html/MEASUR11.html

追　記

本稿で紹介した微分幾何の形式証明に限らずどの分野についても MML 作成に参加ご協力をお願いしたい。ご協力いただいた成果は論文誌 Formalized Mathematics に投稿可能である。

(Formalized Mathematics は zbMATH や Web of Science "Emerging Sources Citation Index"にインデックスされている)

論文内容の新規性は数学としての新規性ではなくて形式化としての新規性であり、古典的・初等的なものでも今までの MML になければ受け付けられる。　既にあるものでも発展させていれば歓迎される。　参加をご検討頂ければ幸いである。